

Five and Seven Qubit Codes

5-Qubit code

In Chapter 9 of the text we introduced the 9-qubit Shor code. It is capable of diagnosing and correcting all single-qubit errors. As the first quantum error correcting code (QECC), the Shor code is of historical significance. But, soon after its introduction, 5-qubit and 7-qubit QECC schemes were introduced. We review them below by applying the stabilizer formalism discussed in the text. First, let's consider the 5-qubit code.

We posit the set of 5-qubit stabilizer operators

$$M_0 = Z \otimes X \otimes X \otimes Z \otimes \mathbb{1}$$

$$M_1 = X \otimes X \otimes Z \otimes \mathbb{1} \otimes Z$$

$$M_2 = X \otimes Z \otimes \mathbb{1} \otimes Z \otimes X$$

$$M_3 = Z \otimes \mathbb{1} \otimes Z \otimes X \otimes X$$

```
In[1]:= (* Define the Pauli operators, and the single qubit identity *)
Z = {{1, 0}, {0, -1}};
X = {{0, 1}, {1, 0}};
Y = {{0, -I}, {I, 0}};
unit = {{1, 0}, {0, 1}};
(* Hadamard Gate *)
H = 1 / Sqrt[2] {{1, 1}, {1, -1}};
```

Using these gates we construct the stabilizers

```
In[6]:= M0 = KroneckerProduct[Z, X, X, Z, unit];
M1 = KroneckerProduct[X, X, Z, unit, Z];
M2 = KroneckerProduct[X, Z, unit, Z, X];
M3 = KroneckerProduct[Z, unit, Z, X, X];
```

Exercise 1. What are the matrix dimensions of these operators ?

Exercise 2. Using the above matrix representations, confirm that the stabilizers are mutually commuting, i.e. $[M_i, M_j] = 0$ for all combinations of i, j .

From our discussion in the text of the stabilizer formalism, we know that the logical codewords are eigenstates of the stabilizers with eigenvalue +1, i.e.

$$M_i|0\rangle_{5L} = |0\rangle_{5L}$$

$$M_i|1\rangle_{5L} = |1\rangle_{5L}$$

where $|0\rangle_{5L}$, $|1\rangle_{5L}$ are 5-qubit codewords for the binary logical values 0, 1 respectively.

Let's define the following operators

```
In[10]:= (* Unit is the 5-qubit identity operator *)
Unit = IdentityMatrix[2^5];
projector = 1 / 4 (Unit + M0) . (Unit + M1) . (Unit + M2) . (Unit + M3);
```

Exercise 3. Is operator projector really a projection operator ? (Hint: a projection operator P has the property $P \cdot P = P$)

We claim that the action of operator projector on the states $|00000\rangle$ and $|11111\rangle$ results in the codewords, $|0\rangle_{5L}$, $|1\rangle_{5L}$ respectively, i.e

$$|0\rangle_{5L} = \frac{1}{4}(\mathbb{1} + M_0)(\mathbb{1} + M_1)(\mathbb{1} + M_2)(\mathbb{1} + M_3)|00000\rangle$$

$$|1\rangle_{5L} = \frac{1}{4}(\mathbb{1} + M_0)(\mathbb{1} + M_1)(\mathbb{1} + M_2)(\mathbb{1} + M_3)|11111\rangle$$

Or,

```
In[12]:= (* define single qubit states *)
ket0 = {1, 0};
ket1 = {0, 1};
codeword0 = projector.Flatten[KroneckerProduct[ket0, ket0, ket0, ket0, ket0]]
codeword1 = projector.Flatten[KroneckerProduct[ket1, ket1, ket1, ket1, ket1]]
```

```
Out[14]= { 1/4, 0, 0, 1/4, 0, -1/4, 1/4, 0, 0, -1/4, -1/4, 0, 1/4, 0, 0,
-1/4, 0, 1/4, -1/4, 0, -1/4, 0, 0, -1/4, 1/4, 0, 0, -1/4, 0, -1/4, -1/4, 0 }
```


```
Out[15]= { 0, -1/4, -1/4, 0, -1/4, 0, 0, 1/4, -1/4, 0, 0, -1/4, 0, -1/4,
1/4, 0, -1/4, 0, 0, 1/4, 0, -1/4, -1/4, 0, 0, 1/4, -1/4, 0, 1/4, 0, 0, 1/4 }
```

Exercise 4. Re-express the states codeword0, codeword1, as a linear combination of the basis kets $|i\rangle_5$.

Exercise 5. Using the matrix representations for the stabilizers M_i , and the states codeword0, codeword1, demonstrate that the latter are indeed eigenstates of all M_i with eigenvalue +1. Show that these codewords are orthonormal.

5-Qubit Syndrome Circuit

In the figure below we illustrate the series of gates that constitutes the syndrome measurements for this QECC. The first five wires (starting from the bottom of the figure) represent the 5-qubit register, whereas the upper 4 wires represent the ancillary qubits, initially set to the state $|0000\rangle$.



(1)

In[16]:=

```
(* define ancillary qubit register set to state |0000> *)
ancillary = Flatten[KroneckerProduct[ket0, ket0, ket0, ket0]]
```

Out[16]= {1, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}

In[17]:=

```
(* gate0 represent the first 4 Hadamard gates acting on the ancillary
qubit register, and its direct product with the 5-qubit register *)
gate0 = KroneckerProduct[H, H, H, H, unit, unit, unit, unit, unit];
```

In[18]:=

```
(* gate1 represents the first cluster of control-
gates surrounded by the dotted line in the figure *)
gate1 = KroneckerProduct[unit, unit, unit, {{1, 0}}, {{0, 0}}, unit, unit, unit, unit,
unit] + KroneckerProduct[unit, unit, unit, {{0, 0}}, {{0, 1}}, Z, X, X, Z, unit];
```

In[19]:=

```
(* gates2, gate3, gate4,
correspond to the remaining three cluster of control gates in the figure*)
gate2 =
KroneckerProduct[unit, unit, {{1, 0}}, {{0, 0}}, unit, unit, unit, unit, unit, unit] +
KroneckerProduct[unit, unit, {{0, 0}}, {{0, 1}}, unit, X, X, Z, unit, Z];
```

In[20]:=

```
gate3 =
KroneckerProduct[unit, {{1, 0}}, {{0, 0}}, unit, unit, unit, unit, unit, unit, unit] +
KroneckerProduct[unit, {{0, 0}}, {{0, 1}}, unit, unit, X, Z, unit, Z, X];
```

In[21]:=

```
gate4 =
KroneckerProduct[{{1, 0}}, {{0, 0}}, unit, unit, unit, unit, unit, unit, unit, unit] +
KroneckerProduct[{{0, 0}}, {{0, 1}}, unit, unit, unit, Z, unit, Z, X, X];
```

```
In[22]:= (* Syndrome circuit *)
syndrome = gate0.gate4.gate3.gate2.gate1.gate0;
```

Syndrome Measurements


```
In[23]:= (* construct total 4+5 qubit state for codewords *)
state0 = Flatten[KroneckerProduct[ket0, ket0, ket0, ket0, codeword0]];
state1 = Flatten[KroneckerProduct[ket0, ket0, ket0, ket0, codeword1]];

```

Exercise 6. Is operator syndrome Hermitian ?

Exercise 7. Demonstrate that the codewords, state0, state1, are +1 eigenstates of syndrome

In order to analyze the action of the syndrome on the codeword, let's look at a simpler circuit shown by the diagram below



(2)

In this figure the shaded area represent stabilizer gate M_0 sandwiched by two Hadamard gates. We included an X-type (i.e. bit flip) error in the wire of the second qubit. So if we focus on the first 5-qubit subspace and assume that the input state is the codeword $|0\rangle_{5L}$, this error gate results in the vector $EX_2|0\rangle_{5L}$

where error gate,

$$EX_2 \equiv \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \otimes X \otimes \mathbb{1}$$

Including the ancillary qubit, initially in state $|0\rangle$, the circuit results in the following output

$$(H \otimes \mathbb{1}^{\otimes 5}) cM_0 (H|0\rangle \otimes EX_2|0\rangle_{5L})$$

where cM_0 represents the control gate connecting the ancillary control qubit with the target qubits 1-5. Now using the properties of the Hadamard and control gates we get,

$$\begin{aligned} & \sqrt{2}(H \otimes \mathbb{1}^{\otimes 5}) cM_0 (|0\rangle + |1\rangle) \otimes EX_2|0\rangle_{5L} = \\ & \sqrt{2}(H \otimes \mathbb{1}^{\otimes 5}) |0\rangle \otimes EX_2|0\rangle_{5L} + \sqrt{2}(H \otimes \mathbb{1}^{\otimes 5}) |1\rangle \otimes M_0 EX_2|0\rangle_{5L} \end{aligned}$$

Exercise 8. Prove that $M_0 EX_2 |0\rangle_{5L} = -EX_2 |0\rangle_{5L}$.

Using this relation we find, that Eq. (3) reduces to

$$\frac{1}{2}(|0\rangle + |1\rangle) \otimes EX_2|0\rangle_{5L} - \frac{1}{2}(|0\rangle - |1\rangle) \otimes EX_2|0\rangle_{5L} = |1\rangle \otimes EX_2|0\rangle_{5L}$$

Exercise 9. If instead of the error considered above, the system is subjected to the errors boxes

$$\begin{aligned} EX_3 &= \mathbb{1} \otimes \mathbb{1} \otimes X \otimes \mathbb{1} \otimes \mathbb{1} \\ EX_4 &= \mathbb{1} \otimes X \otimes \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \\ EX_1 &= \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \otimes X \\ E_0 &= \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \otimes \mathbb{1} \end{aligned}$$

show that the output of the circuit is

$$\frac{1}{2}(|0\rangle + |1\rangle) \otimes EX_i|0\rangle_{5L} + \frac{1}{2}(|0\rangle - |1\rangle) \otimes EX_i|0\rangle_{5L} = |0\rangle \otimes EX_i|0\rangle_{5L}$$

We can now apply this analysis to the full syndrome 5 + 4 qubit circuit shown in Fig. (1). Because of the identity $HH = 1$, we insert a pair of Hadamard gates between each cluster of stabilizers M_j . Using the results of the analysis given above for Fig. (2), convince yourself

Exercise 10. That the action of error EX_2 in the first cluster of stabilizer M_0 (outlined by the dashed line), leads to the following output

$$|0001\rangle \otimes EX_2|0\rangle_{5L}$$

As this state evolves, it meets the subsequent stabilizers M_1, M_2, M_3 . We repeat this analysis to arrive at the final output of the syndrome circuit

Exercise 11. Show the action of error EX_2 leads to the following output in the syndrome circuit shown in

Fig. (1)

$$|0101\rangle \otimes EX_2|0\rangle_{5L}$$

Let's check this result by simulating the syndrome circuit in Fig. (1). First, we construct the error box for EX_2 .


```
In[27]:= (* note:
```

here EX2 is written as direct product with a unit ancillary qubit operator)*)

```
guess = EX2.Flatten[KroneckerProduct[ket0, ket1, ket0, ket1, codeword0]]
```

[illegible]

```
In[28]:= output == guess
```

Out[28]= True

So indeed our simulation is in harmony with the expected answer.

Problem1. Calculate the syndrome assignments of all possible single- X_i type errors.

Problem2. Calculate the syndrome assignments of all possible single- Z_i type errors.

Problem3. Calculate the syndrome assignments of all possible single- Y_i type errors

From the results of Problems 1,2,3 we find that each single-qubit error has a unique syndrome signature stored in the ancillary qubits. In general, the output has the form

$$|i\rangle_4 \otimes E_i |0\rangle_{5L}$$

where E_i is one of the error gates, and index i is the unique tag for that error. If a measurement of the ancillary qubits reveals index k , the subsequent action of gate E_k , as it is a member of the Pauli group

